


Stata Graphics



Stata Graphics Syntax

`graph <graphtype>`

`graph bar`

`graph twoway <plottype>`

`graph twoway scatter`

`graph twoway line`

`graph twoway lfit`

`graph twoway lfitci`

graphs commands may have options

some options have suboptions or a list of options

`graph twoway scatter var1 var2, xlabel(30(10)100, labsize(small))`

appearance of graph defined by graph elements:

data - marker symbols, lines


elements within plot region – text, marker labels, line labels

elements outside plot region – titles, legend, notes, axis labels, tick marks, axis titles

size and shape of plot region and entire graph


Intro To Graphing In Stata

```
. sysuse auto, clear  
. graph twoway scatter mpg weight
```



“graph” is often optional. So is
“twoway” in this case.


```
. scatter mpg weight
```



Note: Nearly all graphing commands start with “graph”, and “twoway” is a large family of graphs.

Creating Multiple Graphs with “by():”


```
. twoway scatter mpg weight, by(foreign)
```



Note that the value label is displayed above the graphs, and the variable label is displayed in the bottom right hand corner.


Overlaying “twoway” graphs

```
. twoway scatter mpg weight || lfit mpg weight
```



The || tells Stata to put the second graph on top of the first one – order matters! You don’t need to type “twoway” twice; it applies to both.


```
. twoway (scatter mpg weight) (lfit mpg weight)
```



This is another way of writing the command – it doesn’t matter which one you use.


Twoway density


- Syntax
 - `graph twoway (plot1, opts) (plot2, opts), opts`
- One plot
 - `kdensity Duration`
- Two plots overlaid
 - `twoway ( kdensity Duration if Group ==0, lcolor(blue) )`
`( kdensity Duration if Group ==1, lcolor(red) )`
- Side *by* side
 - `twoway ( kdensity Duration), by(Group)`



twoway (kdensity Duration if Group ==0,
lcolor(blue)) (kdensity Duration if Group
==1, lcolor(red))

kdensity Duration





twoway (kdensity Duration),
by(Group)

Stata Graphics Syntax: A Simple Example

```
sysuse uslifeexp.dta, clear
```


```
graph twoway line le year
```

```
/* OR */
```

```
twoway line le year
```

```
/* OR */
```

```
line le year
```




Using Schemes

```
line le year, scheme(slmono)
```




```
line le year, scheme(economist)
```

```
/* to see list of  
scheme names:  
graph query, schemes  
  
to change default scheme:  
set scheme schemename  
*/
```




Multiple Dependent Variables

```
line le_wmale le_wfemale le_bmale le_bfemale year
```



Adding Text

```
line le_wmale le_wfemale le_bmale le_bfemale year ///  
, text(32 1920 "{bf:1918} {it:Influenza} Pandemic", place(3))
```



Overlaying Two-Way Plot Types

```
scatter le year if year >= 1950 || lfit le year if year >= 1950
```

```
/* OR */
```


```
scatter ///  
le year if year >= 1950 ///  
|| lfit le year if year >= 1950
```

```
/* OR */
```

```
twoway ///  
(scatter le year if year >= 1950) ///  
(lfit le year if year >= 1950)
```


```
/* OR */
```

```
#delimit ;  
twoway  
(scatter le year if year >= 1950)  
(lfit le year if year >= 1950);  
#delimit cr
```




Overlaying Two-Way Plot Types

```
scatter le year if year >= 1925    ///  
|| lfit le year if year >= 1925 & ///  
    year < 1950    ///  
|| lfit le year if year >= 1950  
  
    /* OR */  
twoway                                ///  
(scatter le year if year >= 1925)    ///  
(lfit    le year if year >= 1925 & ///  
    year < 1950)    ///  
(lfit    le year if year >= 1950)  
  
    /* OR */  
  
#delimit ;  
scatter le year if year >= 1925  
|| lfit le year if year >= 1925 & year < 1950  
|| lfit le year if year >= 1950;  
#delimit cr
```




Overlaying Two-Way Plot Types

```
#delimit ;  
scatter le_male le_female year if year >= 1950  
|| lfit le_male year if year >= 1950  
|| lfit le_female year if year >= 1950;  
#delimit cr
```




Adding a Title and Removing the Legend

```
#delimit ;  
scatter le_male le_female year if year >= 1950  
|| lfit le_male year if year >= 1950  
|| lfit le_female year if year >= 1950  
,title("US Male and Female Life Expectancy, 1950-2000")  
text(75 1978 "Female", place(3))  
text(68 1978 "Male", place(3))  
legend(off);  
#delimit cr
```




Showing Confidence Intervals, Labelling Axes, Modifying Legend

```
sysuse lifeexp.dta, clear
#delimit ;
twoway
  (lfitci lexp safewater if region == 2) /* North America */
  (scatter lexp safewater if region == 2)
, title("Life expectancy at birth by access to safe water, 1998")
  ytitle("Life expectancy at birth")
  xtitle("Percent of population with access to safe water")
  legend(ring(0) pos(5) order(2 "Linear fit" 1 "95% CI"));
#delimit cr
```




Markers Labels and Subtitles

```
twoway
  (lfitci lexp safewater if region == 2) /* North America */
  (scatter lexp safewater if region == 2, mlabel(country))
, title("Life expectancy at birth by access to safe water, 1998")
  subtitle("North America")
  ytitle("Life expectancy at birth")
  xtitle("Percent of population with access to safe water")
  legend(ring(0) pos(5) order(2 "Linear fit" 1 "95% CI"));
```




By-Graph: Separate Graphs for Each Subset of Data

```
#delimit ;  
twoway scatter lexp safewater, by(region, total)  
, ytitle("Life expectancy at birth")  
  xtitle("Percent of population with access to safe  
water");  
#delimit cr
```




By-Graph Options

```
twoway scatter lexp safewater  
,by(region,total style(compact)  
    title("Life expectancy by access to safe water") note(""))  
ytitle("Life expectancy at birth")  
xtitle("Percent of population with access to safe water");
```




Axis Scale, Ticks and Labels

```
twoway scatter lexp safewater
, by(region,total style(compact)
    title("Life expectancy by access to safe water") note(""))
xscale(range(20 100))
xtick(20(10)100)
xlabel(30(10)100, labsize(small))
xtitle("Percent of population with access to safe water")
ytitle("Life expectancy at birth")
ylabel(55(5)80, angle(0));
```




Combining Graphs

```
graph combine north_america south_america  
,title("Life expectancy by access to safe water", color(black)) col(1);
```



More Graphs


Graph matrix is a great alternative to a correlation matrix to investigate relationships between variables




Example using gestational data

Univariate

- Density
 - `kdensity weight`




- Boxplot
 - `graph hbox weight`



Bivariate

- Scatter
 - scatter weight gest




Scatter and density plots for many types of data


Y-type	X-type	Scatter	Density
Cont			x
Cont	Cont	x	
Cont	Cat	x	x
Binary	Cont	x	

Continuous by 3 categories

- Is birth weight the same over parity?




Equal means? Linear effect?
Outliers?



Equal variances?

Continuous by 3 categories

Scatterplot




twoway (scatter weight parity3)
(fpfitci weight parity3) (lfit weight
parity3), legend(off)

- Look for:
 - Outliers (all analyses)
 - Non-linear effects (regression)

Continuous by continuous

- twoway
 (scatter weight gest)
 (fpfitci weight gest)
 (lfit weight gest)

- Look for:
 - Main effect (line)
 - Non-linearity (smooth)
 - outliers



Saving and Including Stata Graphs

save graph in portable format (format determined by filename extension)

vector formats contain drawing instructions (.wmf .emf .ps .eps .pdf)
resolution independent
work well if graph may be resized

```
graph export north_america.wmf
```

raster formats save graph pixel-by-pixel (.png)
use current resolution
work well if including graph on web pages

```
graph export north_america.png
```

include "portable-format-graph" in Windows application (Word, Powerpoint):

Insert -> Picture -> From File

Useful Stata graph websites: <https://www.ssc.wisc.edu/sscc/pubs/4-24.htm>
<https://www.stata.com/support/faqs/graphics/gph/stata-graphs/>